
====FIR_brickwall_SquareWave_DC_Controlled====

FIR_brickwall
*  ___    _____  V1  _____ V2   _____
* |VIN|__| 1/z |____| 1/z |____| 1/z |_....
* |___|  |_____|  | |_____| |  |_____|
*               __|__     __|__
*              |  a1 |   |  a2 |       NORMAL FIR FORMAT
*              |_____|   |_____|
*                 |         |      ___
*                 |         |___\ / _ \      ___
*                 |             // \   \____|OUT|
*                 |_____________\\ /_  /    |___|
*                               / \___/
*..............................................................
*
*         F1        F2       F3        F4
* ___    __|__ V1  __|__ V2 __|__ V3  __|__  V4
*|VIN|__|D_LAY|___|D_LAY|__|D_LAY|___|D_LAY|_...
*|___|  |_____|   |_____|  |_____|   |_____|
*          |         |        |         |
*          |_________|________|_________|...
*                   |
*                   |  ____   DC VOLTAGE CONTROLED FORMAT
*       _/\  /\  /\_|_|VOUT|
*     _|_  \/  \/     |____|
*     ///   R1 = 1 Ohms
*

The sanity check for this FIR digital filter is how it responds to
a square wave. The response should match what happens when a square wave
is brickwalled filtered using FFT methods. 

The output should resemble the square wave in terms of its fundamental
and all its harmonics where in this case any harmonic above the 5th
has been removed. Equally important is that none of the remaining
harmonics should have any shift in time or phase compared to the
fundamental. From the point of view of the 16th delay point,
this is how things appear to be. 

*=========Input_Signals===========================================
*V_PULSE  NODE_P NODE_N DC     VALUE  PULSE( VINIT  VPULSE TDELAY TRISE  TFALL  PWIDTH PERIOD )
VIN       VIN    0      DC     0      PULSE( 0      1      10u    10u    10u     .5   1 ) 
VT        VT     0      DC     0      PWL(   0      0      100    100)
VFIN      FIN    0      DC     1
VFC       FC     0      DC     1
VFS       FS     0      DC     50
BVCLK     CLK    0      V      =  u(sin(6.283*V(FS)*V(VT)))
BF12      F12A   0      V      =  2*sin(3.14*V(FC)/V(FS))

The impulse has  been converted to a square wave. 



*=========SCALED_DELAY===========================================
XZ_DLAY   VIN    VINZ   CLK              Z_DLAY
XS_DLAY1  VINZ   V1     VOUT   CLK  F17  S_DLAY
XS_DLAY2  V1     V2     VOUT   CLK  F16  S_DLAY
XS_DLAY3  V2     V3     VOUT   CLK  F15  S_DLAY
XS_DLAY4  V3     V4     VOUT   CLK  F14  S_DLAY
XS_DLAY5  V4     V5     VOUT   CLK  F13  S_DLAY
XS_DLAY6  V5     V6     VOUT   CLK  F12  S_DLAY
XS_DLAY7  V6     V7     VOUT   CLK  F11  S_DLAY
XS_DLAY8  V7     V8     VOUT   CLK  F10  S_DLAY
XS_DLAY9  V8     V9     VOUT   CLK  F9   S_DLAY
XS_DLAY10 V9     V10    VOUT   CLK  F8   S_DLAY
XS_DLAY11 V10    V11    VOUT   CLK  F7   S_DLAY
XS_DLAY12 V11    V12    VOUT   CLK  F6   S_DLAY
XS_DLAY13 V12    V13    VOUT   CLK  F5   S_DLAY
XS_DLAY14 V13    V14    VOUT   CLK  F4   S_DLAY
XS_DLAY15 V14    V15    VOUT   CLK  F3   S_DLAY
XS_DLAY16 V15    V16    VOUT   CLK  F2   S_DLAY
XS_DLAY17 V16    V17    VOUT   CLK  F3   S_DLAY
XS_DLAY18 V17    V18    VOUT   CLK  F4   S_DLAY
XS_DLAY19 V18    V19    VOUT   CLK  F5   S_DLAY
XS_DLAY20 V19    V20    VOUT   CLK  F6   S_DLAY
XS_DLAY21 V20    V21    VOUT   CLK  F7   S_DLAY
XS_DLAY22 V21    V22    VOUT   CLK  F8   S_DLAY
XS_DLAY23 V22    V23    VOUT   CLK  F9   S_DLAY
XS_DLAY24 V23    V24    VOUT   CLK  F10  S_DLAY
XS_DLAY25 V24    V25    VOUT   CLK  F11  S_DLAY
XS_DLAY26 V25    V26    VOUT   CLK  F12  S_DLAY
XS_DLAY27 V26    V27    VOUT   CLK  F13  S_DLAY
XS_DLAY28 V27    V28    VOUT   CLK  F14  S_DLAY
XS_DLAY29 V28    V29    VOUT   CLK  F15  S_DLAY
XS_DLAY30 V29    V30    VOUT   CLK  F16  S_DLAY
XS_DLAY31 V30    V31    VOUT   CLK  F17  S_DLAY
VF1       F1     0      DC     0
VF2       F2     0      DC     +1.000
VF3       F3     0      DC     +0.900
VF4       F4     0      DC     +0.640
VF5       F5     0      DC     +0.300
VF6       F6     0      DC     +0.000
VF7       F7     0      DC     -0.175
VF8       F8     0      DC     -0.207
VF9       F9     0      DC     -0.125
VF10      F10    0      DC     +0.000
VF11      F11    0      DC     +0.096
VF12      F12    0      DC     +0.118
VF13      F13    0      DC     +0.075
VF14      F14    0      DC     +0.000
VF15      F15    0      DC     -0.060
VF16      F16    0      DC     -0.080
VF17      F17    0      DC     -0.060

*TRAN     TSTEP  TSTOP  TSTART TMAX   ?UIC?
.tran     30u    1     0      30u    UIC
.control
run
set       pensize = 2
plot      v16   vout
.endc

Simulate the same. The square wave response and impulse response are
shown together. 

Add enough delay elements, and the square way response will approach perfection.

*
*         F1        F2       F3        F4
* ___    __|__ V1  __|__ V2 __|__ V3  __|__  V4
*|VIN|__|D_LAY|___|D_LAY|__|D_LAY|___|D_LAY|_...
*|___|  |_____|   |_____|  |_____|   |_____|
*          |         |        |         |
*          |_________|________|_________|...
*                   |
*                   |  ____   DC VOLTAGE CONTROLED FORMAT
*       _/\  /\  /\_|_|VOUT|
*     _|_  \/  \/     |____|
*     ///   R1 = 1 Ohms



===========Full_Netlist_For_Copy_Paste=======================
FIR_brickwall_Square
*  ___    _____      _____      _____
* |VIN|__| 1/z |____| 1/z |____| 1/z |_....
* |___|  |_____|  | |_____| |  |_____|
*               __|__     __|__
*              |  a1 |   |  a2 |       NORMAL FIR FORMAT
*              |_____|   |_____|
*                 |         |      ___
*                 |         |___\ / _ \      ___
*                 |             // \   \____|OUT|
*                 |_____________\\ /_  /    |___|
*                               / \___/
*..............................................................
*

*         F1        F2       F3        F4
* ___    __|__ V1  __|__ V2 __|__ V3  __|__  V4
*|VIN|__|D_LAY|___|D_LAY|__|D_LAY|___|D_LAY|_...
*|___|  |_____|   |_____|  |_____|   |_____|
*          |         |        |         |
*          |_________|________|_________|
*                   |
*                   |  ____   DC VOLTAGE CONTROLED FORMAT
*       _/\  /\  /\_|_|VOUT|
*     _|_  \/  \/     |____|
*     ///   R1 = 1 Ohm
*

*=========Create_Signal==================
*V_PULSE  NODE_P NODE_N DC     VALUE  PULSE( VINIT  VPULSE TDELAY TRISE  TFALL  PWIDTH PERIOD )
VIN       VIN    0      DC     0      PULSE( 0      1      10u    10u    10u     .5   1 ) 
VT        VT     0      DC     0      PWL(   0      0      100    100)
VFIN      FIN    0      DC     1
VFC       FC     0      DC     1
VFS       FS     0      DC     50

BVCLK     CLK    0      V      =  u(sin(6.283*V(FS)*V(VT)))
BF12      F12A   0      V      =  2*sin(3.14*V(FC)/V(FS))

VF1       F1     0      DC     0
VF2       F2     0      DC     +1.000
VF3       F3     0      DC     +0.900
VF4       F4     0      DC     +0.640
VF5       F5     0      DC     +0.300
VF6       F6     0      DC     +0.000
VF7       F7     0      DC     -0.175
VF8       F8     0      DC     -0.207
VF9       F9     0      DC     -0.125
VF10      F10    0      DC     +0.000
VF11      F11    0      DC     +0.096
VF12      F12    0      DC     +0.118
VF13      F13    0      DC     +0.075
VF14      F14    0      DC     +0.000
VF15      F15    0      DC     -0.060
VF16      F16    0      DC     -0.080
VF17      F17    0      DC     -0.060

XZ_DLAY   VIN    VINZ   CLK              Z_DLAY
XS_DLAY1  VINZ   V1     VOUT   CLK  F17  S_DLAY
XS_DLAY2  V1     V2     VOUT   CLK  F16  S_DLAY
XS_DLAY3  V2     V3     VOUT   CLK  F15  S_DLAY
XS_DLAY4  V3     V4     VOUT   CLK  F14  S_DLAY
XS_DLAY5  V4     V5     VOUT   CLK  F13  S_DLAY
XS_DLAY6  V5     V6     VOUT   CLK  F12  S_DLAY
XS_DLAY7  V6     V7     VOUT   CLK  F11  S_DLAY
XS_DLAY8  V7     V8     VOUT   CLK  F10  S_DLAY
XS_DLAY9  V8     V9     VOUT   CLK  F9   S_DLAY
XS_DLAY10 V9     V10    VOUT   CLK  F8   S_DLAY
XS_DLAY11 V10    V11    VOUT   CLK  F7   S_DLAY
XS_DLAY12 V11    V12    VOUT   CLK  F6   S_DLAY
XS_DLAY13 V12    V13    VOUT   CLK  F5   S_DLAY
XS_DLAY14 V13    V14    VOUT   CLK  F4   S_DLAY
XS_DLAY15 V14    V15    VOUT   CLK  F3   S_DLAY
XS_DLAY16 V15    V16    VOUT   CLK  F2   S_DLAY
XS_DLAY17 V16    V17    VOUT   CLK  F3   S_DLAY
XS_DLAY18 V17    V18    VOUT   CLK  F4   S_DLAY
XS_DLAY19 V18    V19    VOUT   CLK  F5   S_DLAY
XS_DLAY20 V19    V20    VOUT   CLK  F6   S_DLAY
XS_DLAY21 V20    V21    VOUT   CLK  F7   S_DLAY
XS_DLAY22 V21    V22    VOUT   CLK  F8   S_DLAY
XS_DLAY23 V22    V23    VOUT   CLK  F9   S_DLAY
XS_DLAY24 V23    V24    VOUT   CLK  F10  S_DLAY
XS_DLAY25 V24    V25    VOUT   CLK  F11  S_DLAY
XS_DLAY26 V25    V26    VOUT   CLK  F12  S_DLAY
XS_DLAY27 V26    V27    VOUT   CLK  F13  S_DLAY
XS_DLAY28 V27    V28    VOUT   CLK  F14  S_DLAY
XS_DLAY29 V28    V29    VOUT   CLK  F15  S_DLAY
XS_DLAY30 V29    V30    VOUT   CLK  F16  S_DLAY
XS_DLAY31 V30    V31    VOUT   CLK  F17  S_DLAY

R1        VOUT   0      .25

.tran     30u    2      0      30u    UIC

.control
run
*TRAN     TSTEP  TSTOP  TSTART TMAX   ?UIC?
*tran      10u    2      0      10u   



set       pensize = 2
plot      v16 vout

.endc
*=========Switch_Model===========================================

.MODEL    SW      SW(    VT=.5 VH=.1  RON=100m ROFF=100MEG)

*=========SCALED_DELAY===========================================
*            CLK
*
*  ___      __|__               ____
* |VIN|____|D_LAY|_____________|VOUT|
* |___|    |_____| |           |____|
*                  |     | F1
*                  | |\  V      ____
*                  |_|1\/\/\___|IOUT|
*                    | /\/\/   |____|
*                    |/
*

.SUBCKT   S_DLAY VIN    VOUT  IOUT  CLK  F1
 XZ_DLAY  VIN    VOUT   CLK  Z_DLAY
 BOUT     IOUT   0      I  = -V(VOUT)*V(F1)
.ENDS     S_DLAY

*=========Z_DLAY=========================================
*                          _________________
*                          |   ___________  |
*  ___   |\          VBF   |  |  ____     | | ____
* |CLK|__|1\_______________|_/|\_\   \    | |_|   \
* |___|  | / |           VLP  |   |NE \_  |   | PE \_
*        |/  |/\  /\  /\______|___| 2 / | |___| 3  / |
*               \/  \/  _|_      /___/  |     |___/  |
*               RLP     ___                          |
*                       _|_
*                   CLP ///
*
* ___   |\  V1  /   V2       |\  V3  /   V4     |\   ___
*|IN |__|4\____/ ____________|5\____/ __________|6\_|OUT|
*|___|  | / S1  NE  | _|_    | / S2  PE  | _|_  | / |___|
*       |/          | ___    |/          | ___  |/
*        Rshunt1    |  | C1   Rshunt2    |  |C2
*      _/\  /\  /\__| _|_   _/\  /\  /\__| _|_
*    _|_  \/  \/      /// _|_  \/  \/      ///
*    ///                  ///
.SUBCKT   Z_DLAY VIN    OUT    CLK  
 B1       VBF    0      V =    u( v(CLK )-.5 )
 RLP      VBF    VLP    10k
 CLP      VLP    0      50n    IC=0
 BNOR2    NE     0      V =    1-u( u(v(VBF )-.5)+u(.5 -v(VLP ) ) -.1)
 BAND3    PE     0      V =      u( u(v(VBF )-.5)*u(.5 -v(VLP ) ) -.1)
 B4       V1     0      V =    V(VIN)
 S1       V1     V2     NE     0    SW 
 R1       V2     0      100Meg
 C1       V2     0      1u
 B5       V3     0      V =    V(V2)
 S2       V3     V4     PE     0    SW 
 R2       V4     0      100Meg
 C2       V4     0      100u
 B6       OUT    0      V =    V(V4)
.ENDS     Z_DLAY

.end

